

Details of amendments made by ERG to Roche Erlotinib NSCLC maintenance therapy model

All modifications are activated by simple numerical switches (ModA – ModH, taking value 0 for 'inactive').

Modification A: Time Horizon

Range 't_horizon' set to the formula
=IF(ModA=0, 5, 15)

Modification B: Discounting

On sheet 'New Therapy':

Set cell E6 to the formula

=IF(ModB=0, IF(\$B6 > 11, D6/(1+disc_u)^\$B6, D6), D6/(1+'Model Inputs'!\$C\$42)^ A6)

Set cell K6 to the formula

=IF(ModB=0, IF(\$B6 > 11, J6/(1+disc_u)^\$B6, J6), J6/(1+'Model Inputs'!\$C\$42)^ A6)

Set cell L6 to the formula

=IF(ModB=0, IF(\$B6>11, (\$D6/(1+disc_c)^\$B6), \$D6)*c_adm_new*(1)*CHOOSE(tx_dur, 1, Tx_Duration!E6), \$D6/(1+'Model Inputs'!\$C\$41)^\$A6*c_adm_new*(1)*CHOOSE(tx_dur, 1, Tx_Duration!E6))

Set cell M6 to the formula

=IF(ModB=0, (IF(\$B6>11, (\$D6/(1+disc_c)^\$B6), \$D6))*c_new*(1)*CHOOSE(tx_dur, 1, Tx_Duration!E6), \$D6/(1+'Model Inputs'!\$C\$41)^\$A6*c_new*(1)*CHOOSE(tx_dur, 1, Tx_Duration!E6))

Set cell N6 to the formula

=IF(ModB=0, IF(\$B6 > 11, (\$D6/(1+disc_c)^\$B6)*IF(psa=0, c_pfs, pc_pfs), \$D6*IF(psa=0, c_pfs, pc_pfs)), \$D6/(1+'Model Inputs'!\$C\$41)^\$A6*IF(psa=0, c_pfs, pc_pfs))

Set cell Q6 to the formula

=IF(ModB=0, IF(\$B6 > 11, (\$G6/(1+disc_c)^\$B6)*IF(psa=0, cProg, pc_prog), \$G6*IF(psa=0, cProg, pc_prog)), \$G6/(1+'Model Inputs'!\$C\$41)^\$A6*IF(psa=0, cProg, pc_prog))+IF(ModF=0, 0, Y6)

Set cell U6 to the formula

=IF(ModB=0, IF(\$B6>11, (\$D6/(1+disc_u)^\$B6)* IF(psa=0, IF(ModH=0, u_PFS, 0.6732), pu_pfs), \$D6*IF(psa=0, IF(ModH=0, u_PFS, 0.6732), pu_pfs)), \$D6*IF(psa=0, IF(ModH=0, u_PFS, 0.6732), pu_pfs)/(1+'Model Inputs'!\$C\$42)^A6)

Set cell V6 to the formula

=IF(ModB=0, IF(\$B6>11, (\$G6/(1+disc_u)^\$B6)*IF(psa=0, u_prog, pu_prog), \$G6*IF(psa=0, u_prog, pu_prog)), \$G6*IF(psa=0, u_prog, pu_prog)/(1+'Model Inputs'!\$C\$42)^A6)

Copy formulae in E6, K6, L6, M6, N6, Q6, U6, V6 to the rest of columns E, K, L, M, N, Q, U, V respectively

On sheet 'Comparator':

Set cell E6 to the formula

=IF(ModB=0, IF(\$B6 > 11,D6/(1+disc_u)^\$B6,D6), D6/(1+'Model Inputs'!\$C\$42)^A6)

Set cell K6 to the formula

=IF(ModB=0, IF(\$B6 > 11,J6/(1+disc_u)^\$B6,J6), J6/(1+'Model Inputs'!\$C\$42)^A6)

Set cell L6 to the formula

=IF(ModB=0, (IF(\$B6>11,(\$D6/(1+disc_c)^\$B6),\$D6))*c_adm_std*(adm2pfs_com),
\$D6/(1+'Model Inputs'!\$C\$41)^\$A6*c_adm_std*(adm2pfs_com))

Set cell M6 to the formula

=IF(ModB=0, (IF(\$B6>11,(\$D6/(1+disc_c)^\$B6),\$D6))*c_std*(adm2pfs_com),
\$D6/(1+'Model Inputs'!\$C\$41)^\$A6*c_std*(adm2pfs_com))

Set cell N6 to the formula

=IF(ModB=0,
IF(\$B6 >11,(\$D6/(1+disc_c)^\$B6)*IF(psa=0,c_pfs,pc_pfs),\$D6*IF(psa=0,c_pfs,pc_pfs)),
\$D6/(1+'Model Inputs'!\$C\$41)^\$A6*IF(psa=0,c_pfs,pc_pfs))

Set cell Q6 to the formula

=IF(ModB=0,
IF(\$B6 >11,(\$G6/(1+disc_c)^\$B6)*IF(psa=0,cProg2,pc_prog2),
\$G6*IF(psa=0,cProg2,pc_prog2)),
\$G6/(1+'Model Inputs'!\$C\$41)^\$A6*IF(psa=0,cProg2,pc_prog2))+IF(ModF=0,0,Y6)

Set cell U6 to the formula

=IF(ModB=0,
IF(\$B6>11,(\$D6/(1+disc_u)^\$B6)*IF(psa=0,Comparator!\$U\$6,pu_pfs),
\$D6*IF(psa=0,IF(ModH=0,u_PFS,0.6628),pu_pfs)),
\$D6*IF(psa=0,IF(ModH=0,u_PFS,0.6628),pu_pfs))/(1+'Model Inputs'!\$C\$42)^A6)

Set cell V6 to the formula

=IF(ModB=0,
IF(\$B6>11,(\$G6/(1+disc_u)^\$B6)*IF(psa=0,u_prog,pu_prog),
\$G6*IF(psa=0,u_prog,pu_prog)),
\$G6*IF(psa=0,u_prog,pu_prog))/(1+'Model Inputs'!\$C\$42)^ A6)

Copy formulae in E6,K6,L6,M6,N6,Q6,U6,V6 to the rest of columns E,K,L,M,N,Q,U,V respectively

Modification C: Erlotinib cost

The table shown below was constructed within the 'Dose Table' worksheet to estimate the mean cost of erlotinib treatment within the SATURN trial, based on the Kaplan-Meier PFS result at each 30 day time point during the study indicating the proportion of patients who were eligible to receive a new pack of tablets on that day.

Column L is defined by the formula:

=VLOOKUP(J34,\$G\$34:\$H\$151,2) copied down to cell L62

Column M is defined by the formula:

=INT(J34/365.25) copied down to cell M62

Column N is defined by the formula:

=1/(1+'Model Inputs'!\$C\$41)^'Dose Table'!M34 copied down to cell N62

Cell L65 =SUMPRODUCT(K34:K62,L34:L62)

Cell L67 =L65*'Model Inputs'!E11

Cell N67 =SUMPRODUCT(K34:K62,L34:L62,N34:N62)*'Model Inputs'!E11

\$G\$34:\$H\$151 data is taken directly from SATURN K-M PFS output for the erlotinib arm:

The resulting mean is applied directly to cell 'New Therapy'!M3 as follows:

=IF(ModC=0,SUM(M\$6:INDIRECT("M"&(t_horizon*12 + 5)),'Dose Table'!N67)

Modifications D & E: PFS & OS distributions

A new worksheet 'ERG survival' was created to allow substitution of ERG preferred survival profiles for OS and PFS to be substituted into worksheets 'New Therapy' and 'Comparator'. ModD and ModE take values 0 (original logic), 1 (K-M data), or >1 (ERG model).

The new formulae for PFS are as follows:

'New Treatment'!C6

= IF(ModD=0,
CHOOSE(Distn2,EXP(-plnw*\$B6^pgnw), EXP(-plne*\$B6^pgne), (1 / (1 +
plnl*\$B6^pgnl)), IF(\$B6 = 0,1, (1 - NORMDIST(((LN(\$B6) - plnn) /
pgnn),0,1,TRUE))), EXP((plngo/pgngo)*(EXP(pgngo*\$B6)-1)),
IF(\$B6=0,1,IF(IF(psa1=0,d,pd)>0,1-
GAMMADIST(plnga*\$B6^(pdnga),pgnga,1,TRUE),1-(1-
GAMMADIST(plnga*\$B6^(pdnga),pgnga,1,TRUE))))), 'KM PFS'!\$B4),
VLOOKUP(B6,ERG_PFS_Erl,ModD+1))

'Comparator'!C6

= IF(ModD=0,
CHOOSE(Distn2,EXP(-plcw*\$B6^pgcw),EXP(-
plce*\$B6^pgce),(1/(1+plcl*\$B6^pgcl)),IF(\$B6=0,1,(1-NORMDIST(((LN(\$B6)-
plcn)/pgcn),0,1,TRUE))),EXP((plcgo/pgcgo)*(EXP(pgcgo*\$B6)-
1)),IF(\$B6=0,1,IF(IF(psa1=0,d,pd)>0,1-
GAMMADIST(plcga*\$B6^(pdcga),pgcga,1,TRUE),1-(1-
GAMMADIST(plcga*\$B6^(pdcga),pgcga,1,TRUE))))), 'KM PFS'!\$G4),
VLOOKUP(B6,ERG_PFS_Plac,ModD+1))

The new formulae for OS are as follows:

'New Treatment'!C6

= IF(ModE=0,
 CHOOSE(DthRate_Cen, CHOOSE(distn,EXP(-olnw*\$B6^ognw),EXP(-
 olne*\$B6^ogne),(1/(1+olnl*\$B6^ognl)),IF(\$B6=0,1,(1-NORMDIST(((LN(\$B6)-
 olnn)/ognn),0,1,TRUE))),EXP((olngo/ogngo)*(EXP(ogngo*\$B6)-
 1)),IF(\$B6=0,1,IF(IF(psa1=0,od,opd)>0,1-
 GAMMADIST(olnga*\$B6^(odnga),ognga,1,TRUE),1-(1-
 GAMMADIST(olnga*\$B6^(odnga),ognga,1,TRUE))))),'KM OS'!\$G4),
 CHOOSE(distn,EXP(-colnw*\$B6^cognw),EXP(-
 colne*\$B6^cogne),(1/(1+colnl*\$B6^cognl)),IF(\$B6=0,1,(1-NORMDIST(((LN(\$B6)-
 colnn)/cognn),0,1,TRUE))),EXP((colngo/cogngo)*(EXP(cogngo*\$B6)-
 1)),IF(\$B6=0,1,IF(IF(psa1=0,cod,copd)>0,1-
 GAMMADIST(colnga*\$B6^(codnga),cognga,1,TRUE),1-(1-
 GAMMADIST(colnga*\$B6^(codnga),cognga,1,TRUE))))),'KM OS'!\$G4)),
 VLOOKUP(B6,ERG_OS_Erl,ModE+1))

‘Comparator’!C6

= IF(ModE=0,
 CHOOSE(DthRate_Cen, CHOOSE(distn,EXP(-olcw*\$B6^ogcw), EXP(-
 olce*\$B6^ogce),(1 / (1 + olcl*\$B6^ogcl)), IF(\$B6 = 0,1, (1 - NORMDIST(((LN(\$B6)
 - olcn) / ogcn),0,1,TRUE))), EXP((olcgo/ogcgo)*(EXP(ogcgo*\$B6)-1)),
 IF(\$B6=0,1,IF(IF(psa1=0,od,opd)>0,1-
 GAMMADIST(olcga*\$B6^(odcga),ogcga,1,TRUE),1-(1-
 GAMMADIST(olcga*\$B6^(odcga),ogcga,1,TRUE))))), 'KM OS'!\$B4),
 CHOOSE(distn,EXP(-colcw*\$B6^cogcw), EXP(-colce*\$B6^cogce),(1 / (1 +
 colcl*\$B6^cogcl)), IF(\$B6 = 0,1, (1 - NORMDIST(((LN(\$B6) - colcn) /
 cogcn),0,1,TRUE))), EXP((colcgo/cogcgo)*(EXP(cogcgo*\$B6)-1)),
 IF(\$B6=0,1,IF(IF(psa1=0,cod,copd)>0,1-
 GAMMADIST(colcga*\$B6^(codcga),cogcga,1,TRUE),1-(1-
 GAMMADIST(colcga*\$B6^(codcga),cogcga,1,TRUE))))), 'KM OS'!\$B4)),
 VLOOKUP(B6,ERG_OS_Plac,ModE+1))

The lookup ranges in the ‘ERG survival’ worksheet are located at:

ERG_PFS_Erl = B10:D274

ERG_PFS_Plac = G10:I272

ERG_OS_Erl = L10:N316

ERG_OS_Plac = Q10:S331

Part of the worksheet is shown below (*removed- contains CIC information*)

Modification F: Cost of 2nd line CTX

This makes use of estimated 2nd line CTX costs estimated as follows for cell Y6 of the ‘New Therapy’ and ‘Comparator’ worksheets, and copied down the whole column:

$$Y6 = D6/D\$3*'Post-Prog Txs'!\$B\$95*(332/454)/(1+'Model Inputs'!\$C\$41)^A6$$

N.B. 'Post-Prog Txs'!\\$B\\$95 is the cost per patient of docetaxel 2nd line treatment, and (332/454) is the proportion of patients receiving 2nd line CTX

The modification is applied to cell Q6 and copied to the whole column in both ‘New Therapy’ and ‘Comparator’ worksheets as part of the formula shown above for Modification B. This modification also turns off the original costs of 2nd line treatments as shown below in Modification G.

Modification G: Unit costs update

This is implemented in three cells of the ‘Model Inputs’ worksheet:

Cell E29 = IF(ModG=0, 361.44, 411.67)

Cell E31 =IF(ModF=0,
SUM(IF(ModG=0,1088.93,1225.14),'Post-Prog Txs'!R75),
IF(ModG=0,1088.93,1225.14))

Cell E33 =IF(ModF=0,
SUM(IF(ModG=0,1088.93,1225.14),'Post-Prog Txs'!Q75),
IF(ModG=0,1088.93,1225.14))

Modification H: Utility values

This is implemented within the discounting modified formulae in column U of both ‘New Therapy’ and ‘Comparator’ worksheets as shown above.